Let us study the example of a flexible one-link robot following M. Benosman, G. Le Vey, *Accurate trajectory tracking of flexible arm end-point*, 6th IFAC Symposium on Robot Control, Vienna (Austria), 2000, pp. 569-572.

```
> with(Ore_algebra):
> with(OreModules):
```

Define the Weyl algebra $Alg = A_1$, where Dt acts as differential operator w.r.t. the variable t:

```
> Alg := DefineOreAlgebra(diff=[D,t], polynom=[t],
> comm=[M11,M12,M13,M21,M22,M31,M33,K1,K2,L,alpha,beta]):
```

The system matrix is defined by:

```
> R := evalm([[M11*D^2,M12*D^2,M13*D^2,-1],[M21*D^2,M22*D^2+K1,0,0],
> [M31*D^2,0,M33*D^2+K2,0]]);
```

$$R := \begin{bmatrix} M11\,\mathrm{D}^2 & M12\,\mathrm{D}^2 & M13\,\mathrm{D}^2 & -1 \\ M21\,\mathrm{D}^2 & M22\,\mathrm{D}^2 + K1 & 0 & 0 \\ M31\,\mathrm{D}^2 & 0 & M33\,\mathrm{D}^2 + K2 & 0 \end{bmatrix}$$

The corresponding system of ordinary differential equations is then given by

```
> ApplyMatrix(R, [theta(t),q1(t),q2(t),u(t)], Alg)=evalm([[0]$3]);
```

$$\begin{bmatrix} M11\,(\frac{d^2}{dt^2}\,\theta(t)) + M12\,(\frac{d^2}{dt^2}\,\mathrm{q1}(t)) + M13\,(\frac{d^2}{dt^2}\,\mathrm{q2}(t)) - \mathrm{u}(t) \\ M21\,(\frac{d^2}{dt^2}\,\theta(t)) + K1\,\mathrm{q1}(t) + M22\,(\frac{d^2}{dt^2}\,\mathrm{q1}(t)) \\ M31\,(\frac{d^2}{dt^2}\,\theta(t)) + K2\,\mathrm{q2}(t) + M33\,(\frac{d^2}{dt^2}\,\mathrm{q2}(t)) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

where M denotes the inertia matrix of the arm, $K1$ and $K2$ are the equivalent spring constants, θ is the joint variable, $q = (q1,\, q2)^T$ is the vector of modal coordinates and u is the torque applied to the beam hub. Let us check whether or not the system is controllable, parametrizable and flat.

We first introduce the formal adjoint R_adj of R.

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} M11\,\mathrm{D}^2 & M21\,\mathrm{D}^2 & M31\,\mathrm{D}^2 \\ M12\,\mathrm{D}^2 & M22\,\mathrm{D}^2 + K1 & 0 \\ M13\,\mathrm{D}^2 & 0 & M33\,\mathrm{D}^2 + K2 \\ -1 & 0 & 0 \end{bmatrix}$$

The system is controllable if and only if the Alg-module M associated with R is torsion-free, i.e., if and only if the first extension module ext^1 with values in Alg of the Alg-module associated with R_adj is the zero module. Let us check this last point.

```
> st := time(): Ext := Exti(R_adj, Alg, 1); time()-st;
```

$$Ext := \left[\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix},\right.$$

$[-M33\,\mathrm{D}^2\,M21\,M12 - \mathrm{D}^2\,M22\,M31\,M13 + M33\,\mathrm{D}^2\,M22\,M11\,,\ -M33\,M12\,K1\,,$
$-M13\,M22\,K2\,,\ -M33\,M22]$
$[0\,,\ M12\,M31\,K1\,, -K2\,M21\,M12 + K2\,M22\,M11 - M33\,\mathrm{D}^2\,M21\,M12$
$+ M33\,\mathrm{D}^2\,M22\,M11 - \mathrm{D}^2\,M22\,M31\,M13,\ M22\,M31]$
$[0\,, -\mathrm{D}^2\,M22\,M31\,M13 + M33\,\mathrm{D}^2\,M22\,M11 - K1\,M31\,M13 - M33\,\mathrm{D}^2\,M21\,M12$
$+ M33\,K1\,M11,\ M13\,M21\,K2\,,\ M33\,M21],$
$[M33\,K1\,\mathrm{D}^2 + M33\,\mathrm{D}^4\,M22 + K2\,\mathrm{D}^2\,M22 + K1\,K2]$
$[-M33\,\mathrm{D}^4\,M21 - K2\,\mathrm{D}^2\,M21]$
$[-K1\,M31\,\mathrm{D}^2 - \mathrm{D}^4\,M22\,M31]$
$[M33\,\mathrm{D}^6\,M22\,M11 + M33\,\mathrm{D}^4\,K1\,M11 - M33\,\mathrm{D}^6\,M21\,M12 + K2\,K1\,M11\,\mathrm{D}^2$
$\left. - \mathrm{D}^4\,K1\,M31\,M13 - \mathrm{D}^6\,M22\,M31\,M13 + K2\,\mathrm{D}^4\,M22\,M11 - K2\,\mathrm{D}^4\,M21\,M12]\right]$

0.459

Therefore, as the first matrix *Ext*[1] is the identity, we obtain that the system is generically controllable, and thus, generically parametrizable. A parametrization is then given by *Ext*[3] and we have:

```
> evalm([[theta(t)],[q1(t)],[q2(t)],[u(t)]])=ApplyMatrix(Ext[3], [xi(t)], Alg);
```

$$\begin{bmatrix} \theta(t) \\ \mathrm{q1}(t) \\ \mathrm{q2}(t) \\ \mathrm{u}(t) \end{bmatrix} =$$

$\left[K1\,K2\,\xi(t) + (M33\,K1 + M22\,K2)\,(\frac{d^2}{dt^2}\,\xi(t)) + M33\,M22\,(\frac{d^4}{dt^4}\,\xi(t))\right]$

$\left[-K2\,M21\,(\frac{d^2}{dt^2}\,\xi(t)) - M33\,M21\,(\frac{d^4}{dt^4}\,\xi(t))\right]$

$\left[-K1\,M31\,(\frac{d^2}{dt^2}\,\xi(t)) - M22\,M31\,(\frac{d^4}{dt^4}\,\xi(t))\right]$

$\left[K2\,K1\,M11\,(\frac{d^2}{dt^2}\,\xi(t))\right.$
$+ (-K1\,M31\,M13 + M33\,K1\,M11 + K2\,M22\,M11 - K2\,M21\,M12)\,(\frac{d^4}{dt^4}\,\xi(t))$
$\left.+ (-M33\,M21\,M12 + M33\,M22\,M11 - M22\,M31\,M13)\,(\frac{d^6}{dt^6}\,\xi(t))\right]$

The fact that the *Alg*-module M associated with R is generically torsion-free implies that M is also projective as *Alg* is a hereditary ring. Projectivity of M can be verified by checking whether or not a right-inverse of R exists.

```
> T := simplify(RightInverse(R, Alg));
```

$T :=$

$\left[0\,,\ -\dfrac{M22^2\,(M33\,\mathrm{D}^2+K2)}{M21\,K1\,(-M33\,K1+M22\,K2)}\,,\ \dfrac{M33^2\,(M22\,\mathrm{D}^2+K1)}{M31\,K2\,(-M33\,K1+M22\,K2)}\right]$

$\left[0\,,\ \dfrac{M22\,K2-M33\,K1+M33\,\mathrm{D}^2\,M22}{K1\,(-M33\,K1+M22\,K2)}\,,\ -\dfrac{M33^2\,\mathrm{D}^2\,M21}{M31\,K2\,(-M33\,K1+M22\,K2)}\right]$

$\left[0\,,\ \dfrac{M22^2\,\mathrm{D}^2\,M31}{M21\,K1\,(-M33\,K1+M22\,K2)}\,,\ \dfrac{-M33\,\mathrm{D}^2\,M22-M33\,K1+M22\,K2}{K2\,(-M33\,K1+M22\,K2)}\right]$

$[-1\,, -\mathrm{D}^2(-K2\,M22\,M21\,M12+K2\,M22^2\,M11-M13\,M22^2\,M31\,\mathrm{D}^2$
$-M22\,\mathrm{D}^2\,M21\,M12\,M33+M11\,M22^2\,\mathrm{D}^2\,M33+M21\,K1\,M12\,M33)/(M21\,K1$
$(-M33\,K1+M22\,K2)), \mathrm{D}^2(-M33^2\,\mathrm{D}^2\,M21\,M12+M33^2\,\mathrm{D}^2\,M22\,M11$
$-M33\,M22\,M31\,\mathrm{D}^2\,M13+M33^2\,K1\,M11-M33\,K1\,M31\,M13$
$+M22\,M31\,K2\,M13)/(M31\,K2\,(-M33\,K1+M22\,K2))]$

```
> Mult(R, T, Alg);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Moreover, the fact that the system is time-invariant implies that the *Alg*-module M associated with R is generically free. Let us compute a basis of M.

```
> S := simplify(LeftInverse(Ext[3], Alg));
```

$$S := \left[\begin{array}{cccc} \dfrac{1}{K1\,K2} & \dfrac{M22^2}{M21\,K1\,(-M33\,K1+M22\,K2)} & -\dfrac{M33^2}{M31\,K2\,(-M33\,K1+M22\,K2)} & 0 \end{array}\right]$$

Therefore, if $M33\,K1 \neq M22\,K2$, then the *Alg*-module M is free and, equivalently, the corresponding system is flat. Then, a basis of the *Alg*-module M (a *flat output* of the system) is defined by

```
> xi(t)=ApplyMatrix(S, [theta(t),q1(t),q2(t),u(t)], Alg)[1,1];
```

$$\xi(t) = \frac{\theta(t)}{K1\,K2} + \frac{M22^2\,\mathrm{q1}(t)}{M21\,K1\,(-M33\,K1+M22\,K2)} - \frac{M33^2\,\mathrm{q2}(t)}{M31\,K2\,(-M33\,K1+M22\,K2)}$$

where ξ satisfies $(\theta,\ q1,\ q2,\ u)^T = Ext[3]\ \xi$. Let us compute the Brunovský canonical form in the case where $M33\,K1 \neq M22\,K2$.

```
> B := Brunovsky(R, Alg);
```

$B :=$

$$\left[\frac{1}{K1\,K2},\ \frac{M22^2}{M21\,K1\,(-M33\,K1+M22\,K2)},\ -\frac{M33^2}{M31\,K2\,(-M33\,K1+M22\,K2)},\ 0\right]$$

$$\left[\frac{\mathrm{D}}{K1\,K2},\ \frac{\mathrm{D}\,M22^2}{M21\,K1\,(-M33\,K1+M22\,K2)},\ -\frac{\mathrm{D}\,M33^2}{M31\,K2\,(-M33\,K1+M22\,K2)},\ 0\right]$$

$$\left[0,\ -\frac{M22}{M21\,(-M33\,K1+M22\,K2)},\ \frac{M33}{M31\,(-M33\,K1+M22\,K2)},\ 0\right]$$

$$\left[0,\ -\frac{\mathrm{D}\,M22}{M21\,(-M33\,K1+M22\,K2)},\ \frac{M33\,\mathrm{D}}{M31\,(-M33\,K1+M22\,K2)},\ 0\right]$$

$$\left[0,\ \frac{K1}{M21\,(-M33\,K1+M22\,K2)},\ -\frac{K2}{M31\,(-M33\,K1+M22\,K2)},\ 0\right]$$

$$\left[0,\ \frac{K1\,\mathrm{D}}{M21\,(-M33\,K1+M22\,K2)},\ -\frac{K2\,\mathrm{D}}{M31\,(-M33\,K1+M22\,K2)},\ 0\right]$$

$$\left[0,\ \frac{K1\,(-M33\,K1\,M11+K2\,M21\,M12+K1\,M31\,M13)}{M21\,\%1\,(-M33\,K1+M22\,K2)},\right.$$
$$\left.\frac{K2\,(-K2\,M21\,M12+K2\,M22\,M11-K1\,M31\,M13)}{M31\,\%1\,(-M33\,K1+M22\,K2)},\ \frac{1}{\%1}\right]$$

$$\%1 := M11\,M22\,M33 - M13\,M22\,M31 - M12\,M21\,M33$$

Equivalently, we obtain the following transformation between the Brunovský variables $x[i]$, $i = 1, ..., 6$, v and the system variables θ, $q1$, $q2$ and u.

```
> evalm([seq([x[i](t)],i=1..6),[v(t)]])=ApplyMatrix(B,
> [theta(t),q1(t),q2(t),u(t)], Alg);
```

$$\begin{bmatrix} x_1(t) \\ x_2(t) \\ x_3(t) \\ x_4(t) \\ x_5(t) \\ x_6(t) \\ \mathrm{v}(t) \end{bmatrix} =$$

$$\left[\frac{\theta(t)}{K1\ K2} + \frac{M22^2\ \mathrm{q1}(t)}{M21\ K1\ (-M33\ K1 + M22\ K2)} - \frac{M33^2\ \mathrm{q2}(t)}{M31\ K2\ (-M33\ K1 + M22\ K2)}\right]$$

$$\left[\frac{\frac{d}{dt}\theta(t)}{K1\ K2} + \frac{M22^2\ (\frac{d}{dt}\mathrm{q1}(t))}{M21\ K1\ (-M33\ K1 + M22\ K2)} - \frac{M33^2\ (\frac{d}{dt}\mathrm{q2}(t))}{M31\ K2\ (-M33\ K1 + M22\ K2)}\right]$$

$$\left[-\frac{M22\ \mathrm{q1}(t)}{M21\ (-M33\ K1 + M22\ K2)} + \frac{M33\ \mathrm{q2}(t)}{M31\ (-M33\ K1 + M22\ K2)}\right]$$

$$\left[-\frac{M22\ (\frac{d}{dt}\mathrm{q1}(t))}{M21\ (-M33\ K1 + M22\ K2)} + \frac{M33\ (\frac{d}{dt}\mathrm{q2}(t))}{M31\ (-M33\ K1 + M22\ K2)}\right]$$

$$\left[\frac{K1\ \mathrm{q1}(t)}{M21\ (-M33\ K1 + M22\ K2)} - \frac{K2\ \mathrm{q2}(t)}{M31\ (-M33\ K1 + M22\ K2)}\right]$$

$$\left[\frac{K1\ (\frac{d}{dt}\mathrm{q1}(t))}{M21\ (-M33\ K1 + M22\ K2)} - \frac{K2\ (\frac{d}{dt}\mathrm{q2}(t))}{M31\ (-M33\ K1 + M22\ K2)}\right]$$

$$\left[\frac{K1\ (-M33\ K1\ M11 + K2\ M21\ M12 + K1\ M31\ M13)\ \mathrm{q1}(t)}{M21\ \%1\ (-M33\ K1 + M22\ K2)} + \frac{K2\ (-K2\ M21\ M12 + K2\ M22\ M11 - K1\ M31\ M13)\ \mathrm{q2}(t)}{M31\ \%1\ (-M33\ K1 + M22\ K2)} + \frac{\mathrm{u}(t)}{\%1}\right]$$

$$\%1 := M11\ M22\ M33 - M13\ M22\ M31 - M12\ M21\ M33$$

We easily check that, in the Brunovský coordinates, the system becomes the following Brunovský canonical system.

```
> E := Elimination(linalg[stackmatrix](B, R), [theta,q1,q2,u],
> [seq(x[i],i=1..6),v,0,0,0], Alg):
> ApplyMatrix(E[1], [theta(t),q1(t),q2(t),u(t)], Alg)=ApplyMatrix(E[2],
> [seq(x[i](t),i=1..6), v(t)], Alg);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ \mathrm{u}(t) \\ \mathrm{q2}(t) \\ \mathrm{q1}(t) \\ \theta(t) \end{bmatrix} =$$

$$\left[-(\tfrac{d}{dt}\,x_6(t)) + \mathrm{v}(t)\right]$$
$$\left[-(\tfrac{d}{dt}\,x_5(t)) + x_6(t)\right]$$
$$\left[-(\tfrac{d}{dt}\,x_4(t)) + x_5(t)\right]$$
$$\left[-(\tfrac{d}{dt}\,x_3(t)) + x_4(t)\right]$$
$$\left[-(\tfrac{d}{dt}\,x_2(t)) + x_3(t)\right]$$
$$\left[-(\tfrac{d}{dt}\,x_1(t)) + x_2(t)\right]$$
$$\Big[K2\;K1\;M11\,x_3(t) + (M33\;K1\;M11 - K2\;M21\;M12 - K1\;M31\;M13 + K2\;M22\;M11)\,x_5(t) + (M11\;M22\;M33 - M13\;M22\;M31 - M12\;M21\;M33)\,\mathrm{v}(t)\Big]$$
$$\left[-K1\;M31\,x_3(t) - M22\;M31\,x_5(t)\right]$$
$$\left[-M21\;K2\,x_3(t) - M33\;M21\,x_5(t)\right]$$
$$\left[K1\;K2\,x_1(t) + (M33\;K1 + M22\;K2)\,x_3(t) + M33\;M22\,x_5(t)\right]$$

The output is chosen to be the linear position of the end-effector.

```
> Ry := evalm([[L,alpha,beta,0]]);
```

$$Ry := \begin{bmatrix} L & \alpha & \beta & 0 \end{bmatrix}$$

```
> y(t)=ApplyMatrix(Ry, [theta(t),q1(t),q2(t),u(t)], Alg)[1,1];
```

$$\mathrm{y}(t) = L\,\theta(t) + \alpha\,\mathrm{q1}(t) + \beta\,\mathrm{q2}(t)$$

Let us compute the input-output behaviour of the system.

```
> IO := Elimination(linalg[stackmatrix](R, Ry), [theta,q1,q2,u],
> [seq(0,i=1..3),y], Alg, [u]):
> ApplyMatrix(linalg[submatrix](IO[2], 1..1, 1..2), [y(t),u(t)],
> Alg)[1,1]=0;
```

$$-K2\;K1\;M11\,(\tfrac{d^2}{dt^2}\,\mathrm{y}(t)) + (-K2\;M22\;M11 + K2\;M21\;M12 - M33\;K1\;M11 + K1\;M31\;M13)\,(\tfrac{d^4}{dt^4}\,\mathrm{y}(t)) + (-M11\;M22\;M33 + M13\;M22\;M31 + M12\;M21\;M33)\,(\tfrac{d^6}{dt^6}\,\mathrm{y}(t)) + K2\;K1\;L\,\mathrm{u}(t) + (L\,M33\;K1 + L\,K2\;M22 - K2\;M21\,\alpha - M31\;K1\,\beta)\,(\tfrac{d^2}{dt^2}\,\mathrm{u}(t)) + (-M31\,\beta\,M22 + L\,M33\;M22 - M21\;M33\,\alpha)\,(\tfrac{d^4}{dt^4}\,\mathrm{u}(t)) = 0$$

Let us quickly study the particular case where $M33\;K1 = M22\;K2$. The system matrix becomes:

```
> R_bis := subs(K2=(M33*K1)/M22, evalm(R));
```

$$R\_bis := \begin{bmatrix} M11\,D^2 & M12\,D^2 & M13\,D^2 & -1 \\ M21\,D^2 & M22\,D^2 + K1 & 0 & 0 \\ M31\,D^2 & 0 & M33\,D^2 + \frac{M33\;K1}{M22} & 0 \end{bmatrix}$$

The formal adjoint *R_bisadj* of *R_bis* is defined by:

```
> R_bisadj := Involution(R_bis, Alg);
```

$$R\_bisadj := \begin{bmatrix} M11\,D^2 & M21\,D^2 & M31\,D^2 \\ M12\,D^2 & M22\,D^2 + K1 & 0 \\ M13\,D^2 & 0 & M33\,D^2 + \frac{M33\;K1}{M22} \\ -1 & 0 & 0 \end{bmatrix}$$

The previous study shows that the *Alg*-module *M_bis* associated with *R_bis* contains some torsion elements, and thus, the corresponding system has some autonomous elements. Let us check these results.

```
> TorsionElements(R_bis, [theta(t),q1(t),q2(t),u(t)], Alg);
```

$$\left[\begin{bmatrix} K1\;\theta_1(t) + M22\,(\frac{d^2}{dt^2}\,\theta_1(t)) = 0 \\ K1\;\theta_2(t) + M22\,(\frac{d^2}{dt^2}\,\theta_2(t)) = 0 \\ K1\;\theta_3(t) + M22\,(\frac{d^2}{dt^2}\,\theta_3(t)) = 0 \end{bmatrix},\right.$$

$$\Big[\theta_1(t) =$$

$$(M11\;M22^2\;M33\;M31 - M13\;M22^2\;M31^2 - M31\;M22\;M12\;M21\;M33)\,(\frac{d^2}{dt^2}\,\theta(t))$$

$$+ (-M13\;M22\;M31\;M33\;K1 - M12\;M21\;M33^2\;K1)\,\mathrm{q2}(t) - M22^2\;M33\;M31\;\mathrm{u}(t)\Big]$$

$$[\theta_2(t) = M22\;M31\;\mathrm{q1}(t) - M33\;M21\;\mathrm{q2}(t)]$$

$$\Big[\theta_3(t) = M33\;K1\;M11\;\mathrm{q2}(t)$$

$$\left. + (M11\;M22\;M33 - M13\;M22\;M31 - M12\;M21\;M33)\,(\frac{d^2}{dt^2}\,\mathrm{q2}(t)) + M22\;M31\;\mathrm{u}(t)\Big]\right]$$

```
> Auto := AutonomousElements(R_bis, [theta(t),q1(t),q2(t),u(t)], Alg);
```

$$Auto := \left[\left[\begin{array}{c} \theta_1(t) + M33\ M22\ \theta_3(t) = 0 \\ M12\ K1\ \theta_2(t) + M22\ \theta_3(t) = 0 \\ K1\ \theta_3(t) + M22\ (\frac{d^2}{dt^2}\ \theta_3(t)) = 0 \end{array} \right], \left[\begin{array}{c} \theta_1 = -M33\ M22\ \%1 \\ \theta_2 = -\frac{M22\ \%1}{M12\ K1} \\ \theta_3 = \%1 \end{array} \right], \right.$$

$$\left[\theta_1 = \right.$$

$$(M11\ M22^2\ M33\ M31 - M13\ M22^2\ M31^2 - M31\ M22\ M12\ M21\ M33)\ (\frac{d^2}{dt^2}\ \theta(t))$$

$$\left. + (-M13\ M22\ M31\ M33\ K1 - M12\ M21\ M33^2\ K1)\ \mathrm{q2}(t) - M22^2\ M33\ M31\ \mathrm{u}(t) \right]$$

$$[\theta_2 = M22\ M31\ \mathrm{q1}(t) - M33\ M21\ \mathrm{q2}(t)]$$

$$\left[\theta_3 = M33\ K1\ M11\ \mathrm{q2}(t) \right.$$

$$\left. \left. + (M11\ M22\ M33 - M13\ M22\ M31 - M12\ M21\ M33)\ (\frac{d^2}{dt^2}\ \mathrm{q2}(t)) + M22\ M31\ \mathrm{u}(t) \right] \right]$$

$$\%1 := \_C1\ \sin(\frac{\sqrt{K1}\ t}{\sqrt{M22}}) + \_C2\ \cos(\frac{\sqrt{K1}\ t}{\sqrt{M22}})$$

The second matrix $Auto[2]$ gives the autonomous elements. Moreover, the third matrix $Auto[3]$ gives the expression of the autonomous elements in terms of the system variables θ, $q1$, $q2$ and u whereas the first one $Auto[1]$ corresponds to the system that they satisfy.

Let us recall that the autonomous elements are in one-to-one correspondence with the first integrals of the system. Let us compute the generic first integral of the system.

```
> FirstIntegral(R_bis, [theta(t),q1(t),q2(t),u(t)], Alg);
```

$$-(M22^{(3/2)}\ M31\ (\frac{d}{dt}\ \mathrm{q1}(t))\ \_C1\ \sin(\frac{\sqrt{K1}\ t}{\sqrt{M22}}) + M22^{(3/2)}\ M31\ (\frac{d}{dt}\ \mathrm{q1}(t))\ \_C2\ \cos(\frac{\sqrt{K1}\ t}{\sqrt{M22}})$$

$$- M22\ \mathrm{q1}(t)\ M31\ \_C1\ \cos(\frac{\sqrt{K1}\ t}{\sqrt{M22}})\ \sqrt{K1} + M22\ \mathrm{q1}(t)\ M31\ \_C2\ \sin(\frac{\sqrt{K1}\ t}{\sqrt{M22}})\ \sqrt{K1}$$

$$- M33\ (\frac{d}{dt}\ \mathrm{q2}(t))\ \_C1\ \sin(\frac{\sqrt{K1}\ t}{\sqrt{M22}})\ M21\ \sqrt{M22}$$

$$- M33\ (\frac{d}{dt}\ \mathrm{q2}(t))\ \_C2\ \cos(\frac{\sqrt{K1}\ t}{\sqrt{M22}})\ M21\ \sqrt{M22}$$

$$+ M33\ \mathrm{q2}(t)\ \_C1\ \cos(\frac{\sqrt{K1}\ t}{\sqrt{M22}})\ \sqrt{K1}\ M21 - M33\ \mathrm{q2}(t)\ \_C2\ \sin(\frac{\sqrt{K1}\ t}{\sqrt{M22}})\ \sqrt{K1}\ M21)$$

$$/(M21\ \sqrt{M22})$$

We now consider the same system but with a constant time-delay in the input u due, for instance, to a teleoperation process. We first define the Ore algebra of differential time-delay operators.

```
> Alg2 := DefineOreAlgebra(diff=[D,t], dual_shift=[delta,s], polynom=[t,s],
> comm=[M11,M12,M13,M21,M22,M31,M33,K1,K2], shift_action=[delta,t,h]):
```

The matrix which defines the system is then given by:

```
> R2 := evalm([[M11*D^2,M12*D^2,M13*D^2,-delta],[M21*D^2,M22*D^2+K1,0,0],
> [M31*D^2,0,M33*D^2+K2,0]]);
```

$$R2 := \begin{bmatrix} M11\,D^2 & M12\,D^2 & M13\,D^2 & -\delta \\ M21\,D^2 & M22\,D^2 + K1 & 0 & 0 \\ M31\,D^2 & 0 & M33\,D^2 + K2 & 0 \end{bmatrix}$$

We define the formal adjoint *R2_adj* of *R2* using an involution of *Alg2*.

```
> R2_adj := Involution(R2, Alg2):
```

Let us check whether or not the system is controllable, parametrizable and flat. In order to do that, let us check whether or not the *Alg2*-module associated with *R2* is torsion-free.

```
> Ext1 := Exti(R2_adj, Alg2, 1);
```

$$Ext1 := \left[\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \right.$$
$$[-M33\,D^2\,M21\,M12 - D^2\,M22\,M31\,M13 + M33\,D^2\,M22\,M11\,,\ -M33\,M12\,K1\,,$$
$$-M13\,M22\,K2\,,\ -M33\,\delta\,M22]$$
$$[0\,,\ M12\,M31\,K1\,, -K2\,M21\,M12 + K2\,M22\,M11 - M33\,D^2\,M21\,M12$$
$$+ M33\,D^2\,M22\,M11 - D^2\,M22\,M31\,M13,\ M22\,M31\,\delta]$$
$$[0\,, -D^2\,M22\,M31\,M13 + M33\,D^2\,M22\,M11 - K1\,M31\,M13 - M33\,D^2\,M21\,M12$$
$$+ M33\,K1\,M11,\ M13\,M21\,K2\,,\ M33\,M21\,\delta],$$
$$[M33\,D^4\,\delta\,M22 + M33\,K1\,\delta\,D^2 + K1\,\delta\,K2 + K2\,\delta\,M22\,D^2]$$
$$[-M33\,M21\,\delta\,D^4 - M21\,D^2\,\delta\,K2]$$
$$[-D^4\,M22\,M31\,\delta - \delta\,M31\,K1\,D^2]$$
$$[M33\,D^6\,M22\,M11 + M33\,D^4\,K1\,M11 - M33\,D^6\,M21\,M12 + K2\,K1\,M11\,D^2$$
$$\left. - D^4\,K1\,M31\,M13 - D^6\,M22\,M31\,M13 + K2\,D^4\,M22\,M11 - K2\,D^4\,M21\,M12] \right]$$

As the first matrix *Ext1*[1] is the identity, we deduce that the system is generically controllable and parametrizable. A parametrization is then defined by:

```
> evalm([[theta(t)],[q1(t)],[q2(t)],[u(t)]])=ApplyMatrix(Ext1[3], [xi(t)], Alg2);
```

$$\begin{bmatrix} \theta(t) \\ \mathrm{q1}(t) \\ \mathrm{q2}(t) \\ \mathrm{u}(t) \end{bmatrix} =$$
$$[M33\,M22\,(D^{(4)})(\xi)(t-h) + M33\,K1\,\%1 + K1\,K2\,\xi(t-h) + M22\,K2\,\%1]$$
$$[-M33\,M21\,(D^{(4)})(\xi)(t-h) - M21\,K2\,\%1]$$
$$[-M22\,M31\,(D^{(4)})(\xi)(t-h) - K1\,M31\,\%1]$$
$$[M11\,M22\,M33\,(D^{(6)})(\xi)(t) + M33\,K1\,M11\,(D^{(4)})(\xi)(t)$$
$$- M12\,M21\,M33\,(D^{(6)})(\xi)(t) + K2\,K1\,M11\,(D^{(2)})(\xi)(t)$$
$$- K1\,M31\,M13\,(D^{(4)})(\xi)(t) - M13\,M22\,M31\,(D^{(6)})(\xi)(t)$$
$$+ K2\,M22\,M11\,(D^{(4)})(\xi)(t) - K2\,M21\,M12\,(D^{(4)})(\xi)(t)]$$
$$\%1 := (D^{(2)})(\xi)(t-h)$$

Let us check whether or not the system is flat. In order to do that, we need to check whether or not the *Alg2*-module associated with *R2* is projective.

```
> Ext2 := Exti(R2_adj, Alg2, 2);
```

$$Ext2 := [$$
$$[\delta]$$
$$[M33\,\mathrm{D}^6\,M22\,M11 + M33\,\mathrm{D}^4\,K1\,M11 - M33\,\mathrm{D}^6\,M21\,M12 + K2\,K1\,M11\,\mathrm{D}^2$$
$$- \mathrm{D}^4\,K1\,M31\,M13 - \mathrm{D}^6\,M22\,M31\,M13 + K2\,\mathrm{D}^4\,M22\,M11 - K2\,\mathrm{D}^4\,M21\,M12],$$
$$\left[\ 1\ \right], \mathrm{SURJ}(1)]$$

As the first matrx $Ext2[1]$ is different from the identity matrix, we obtain that the system is not flat. As the torsion-free degree of the *Alg2*-module associated with *R2* is 1, we know that there is a π-polynomial in δ. Let us compute it.

```
> PiPolynomial(R2, Alg2, [delta]);
```

$$[\delta]$$

Therefore, if we allow the time-advance operator δ^{-1} in the system, then the system becomes flat. In particular, a flat output of the system is defined by computing a left-inverse of the parametrization $Ext1[3]$ of the system over $Alg2[\delta^{-1}]$. Let us compute such a left-inverse.

```
> S2 := LocalLeftInverse(Ext1[3], [delta], Alg2);
```

$$S2 := \left[\begin{array}{cccc} \dfrac{1}{\delta\,K1\,K2} & \dfrac{M22^2}{\delta\,M21\,K1\,(-M33\,K1 + M22\,K2)} & -\dfrac{M33^2}{\delta\,M31\,K2\,(-M33\,K1 + M22\,K2)} & 0 \end{array}\right]$$

Then, the flat output of the system is defined by:

```
> psi(t)=ApplyMatrix(S2, [theta(t),q1(t),q2(t),u(t)], Alg2)[1,1];
```

$$\psi(t) = \frac{\theta(t+h)}{K1\,K2} + \frac{M22^2\,\mathrm{q1}(t+h)}{M21\,K1\,(-M33\,K1 + M22\,K2)} - \frac{M33^2\,\mathrm{q2}(t+h)}{M31\,K2\,(-M33\,K1 + M22\,K2)}$$

Let us point out that this flat output corresponds exactly to the advanced flat output of the first model, i.e., without the time-delay in the input of the system.